

Fundamentals of Computer Systems

The MIPS Instruction Set

Stephen A. Edwards

Columbia University

Summer 2020

Instruction Set Architectures

MIPS

The GCD Algorithm

MIPS Registers

Types of Instructions

- Computational

- Load and Store

- Jump and Branch

- Other

Instruction Encoding

- Register-type

- Immediate-type

- Jump-type

Assembler Pseudoinstructions

Higher-Level Constructs

Expressions

Conditionals

Loops

Arrays

Strings & Hello World

ASCII

Subroutines

Towers of Hanoi Example

Factorial Example

Memory Layout

Differences in Other ISAs

Machine, Assembly, and C Code

```
000100001000010100000000000000111  
00000000101001000001000000101010  
00010100010000000000000000000011  
00000000101001000010100000100011  
0000010000000001111111111111100  
00000000100001010010000000100011  
0000010000000001111111111111010  
00000000000001000001000000100001  
00000011111000000000000000001000
```

Machine, Assembly, and C Code

000100001000010100000000000000111	←	beq	\$4, \$5, 28
00000000101001000001000000101010		slt	\$2, \$5, \$4
00010100010000000000000000000011		bne	\$2, \$0, 12
00000000101001000010100000100011		subu	\$5, \$5, \$4
00000100000000011111111111111100		bgez	\$0 -16
0000000010000101001000000100011	←	subu	\$4, \$4, \$5
00000100000000011111111111111010	↳	bgez	\$0 -24
00000000000001000001000000100001		addu	\$2, \$0, \$4
00000011111000000000000000001000		jr	\$31

"Humans"

Machine, Assembly, and C Code

```
000100001000010100000000000000111
00000000101001000001000000101010
00010100010000000000000000000011
00000000101001000010100000100011
0000010000000001111111111111100
00000000100001010010000000100011
0000010000000001111111111111010
00000000000001000001000000100001
00000011111000000000000000001000
```

```
beq $4, $5, 28
slt $2, $5, $4
bne $2, $0, 12
subu $5, $5, $4
bgez $0 -16 = 6
subu $4, $4, $5
bgez $0 -24
→ addu $2, $0, $4
jr $31
```

```
gcd:
    beq $a0, $a1, .L2
    slt $v0, $a1, $a0
    bne $v0, $zero, .L1
    subu $a1, $a1, $a0
    b gcd
.L1:
    subu $a0, $a0, $a1
    b gcd
.L2:
    move $v0, $a0
    j $ra
```

Machine, Assembly, and C Code

```
000100001000010100000000000000111
00000000101001000001000000101010
00010100010000000000000000000011
00000000101001000010100000100011
00000100000000011111111111111100
00000000100001010010000000100011
00000100000000011111111111111010
00000000000001000001000000100001
00000011111000000000000000001000
```

```
beq $4, $5, 28
slt $2, $5, $4
bne $2, $0, 12
subu $5, $5, $4
bgez $0 -16
subu $4, $4, $5
bgez $0 -24
addu $2, $0, $4
jr $31
```

gcd:

```
beq $a0, $a1, .L2
slt $v0, $a1, $a0
bne $v0, $zero, .L1
subu $a1, $a1, $a0
b gcd
.L1:
subu $a0, $a0, $a1
b gcd
.L2:
move $v0, $a0
j $ra
```

```
int gcd(int a, int b)
{
    while (a != b) {
        if (a > b) a = a - b;
        else b = b - a;
    }
    return a;
}
```

← "return"

Algorithms

al·go·rithm

a procedure for solving a mathematical problem (as of finding the greatest common divisor) in a finite number of steps that frequently involves repetition of an operation;
broadly : a step-by-step procedure for solving a problem or accomplishing some end especially by a computer

Merriam-Webster

The Stored-Program Computer

John von Neumann, *First Draft of a Report on the EDVAC*, 1945.

"Since the device is primarily a computer, it will have to perform the ~~elementary~~ operations of arithmetics most frequently. [...] It is therefore reasonable that it should contain *specialized organs for just these operations*.

"If the device is to be [...] as nearly as possible all purpose, then a distinction must be made between the specific instructions given for and defining a particular problem, and the general control organs which see to it that these instructions [...] are carried out. The former must be *stored in some way* [...] the latter are represented by definite operating parts of the device.

"Any device which is to carry out long and complicated sequences of operations (specifically of calculations) *must have a considerable memory*.

Instruction Set Architecture (ISA)



Richard Neutra, Kaufmann House, 1946.

ISA: The interface or **contact** between the hardware and the software


Rules about how to code and interpret machine instructions:

- ▶ Execution model (**program counter**)
- ▶ Operations (**instructions**)
- ▶ Data formats (sizes, addressing modes)
- ▶ Processor state (**registers**)
- ▶ Input and Output (memory, etc.)

Architecture vs. Microarchitecture



Architecture:
The interface the hardware presents to the software



Microarchitecture:
The detailed implementation of the architecture

MIPS


Microprocessor without Interlocked Pipeline Stages

MIPS developed at Stanford by Hennessey et al.

MIPS Computer Systems founded 1984. SGI acquired MIPS in 1992; spun it out in 1998 as MIPS Technologies.

Now, mostly an embedded core competing with ARM.

In many wireless WiFi routers.



RISC vs. CISC Architectures

MIPS is a Reduced Instruction Set Computer. Others include ARM, ~~PowerPC~~, ~~SPARC~~, ~~HP-PA~~, and ~~Alpha~~.

RISC

A Complex Instruction Set Computer (CISC) is one alternative. Intel's x86 is the most prominent example; also Motorola 68000 and DEC VAX.

RISC's underlying principles, due to Hennessy and Patterson:

- ▶ Simplicity favors regularity
- ▶ Make the common case fast
- ▶ Smaller is faster
- ▶ Good design demands good compromises

The GCD Algorithm



Euclid, *Elements*, 300 BC.

The greatest common divisor of two numbers does not change if the smaller is subtracted from the larger.

1. Call the two numbers a and b
2. If a and b are equal, stop: a is the greatest common divisor
3. Subtract the smaller from the larger
4. Repeat steps 2–4

The GCD Algorithm

Let's be a little more explicit:

1. Call the two numbers a and b

2. If a equals b , go to step 8

3. if a is less than b , go to step 6

4. Subtract b from a

5. Go to step 2

6. Subtract a from b

7. Go to step 2

8. Declare a the greatest common divisor

9. Go back to doing whatever you were doing before

$a > b$ here

$a < b$ here

Euclid's Algorithm in MIPS Assembly

gcd:

```
beq $a0, $a1, .L2 # if a = b, go to exit  
sgt $v0, $a1, $a0 # Is b > a?  
bne $v0, $zero, .L1 # Yes, goto .L1
```

```
subu $a0, $a0, $a1 # Subtract b from a (b < a)  
b gcd # and repeat
```

.L1:

```
subu $a1, $a1, $a0 # Subtract a from b (a < b)  
b gcd # and repeat
```

.L2:

```
move $v0, $a0 # return a  
j $ra # Return to caller
```

Instructions

Euclid's Algorithm in MIPS Assembly

gcd:

```
beq $a0, $a1, .L2 # if a = b, go to exit
sgt $v0, $a1, $a0 # Is b > a?
bne $v0, $zero, .L1 # Yes, goto .L1
```

```
subu $a0, $a0, $a1 # Subtract b from a (b < a)
b gcd # and repeat
```

.L1:

```
subu $a1, $a1, $a0 # Subtract a from b (a < b)
b gcd # and repeat
```

.L2:

```
move $v0, $a0 # return a
j $ra # Return to caller
```

Operands: Registers, etc.

Euclid's Algorithm in MIPS Assembly

gcd:

```
beq $a0, $a1, .L2 # if a = b, go to exit
sgt $v0, $a1, $a0 # Is b > a?
bne $v0, $zero, .L1 # Yes, goto .L1

subu $a0, $a0, $a1 # Subtract b from a (b < a)
b gcd # and repeat
```

.L1:

```
subu $a1, $a1, $a0 # Subtract a from b (a < b)
b gcd # and repeat
```

.L2:

```
move $v0, $a0 # return a
j $ra # Return to caller
```

Labels

Euclid's Algorithm in MIPS Assembly

gcd:

```
beq $a0, $a1, .L2 # if a = b, go to exit  
sgt $v0, $a1, $a0 # Is b > a?  
bne $v0, $zero, .L1 # Yes, goto .L1
```

Introduce a comment

```
subu $a0, $a0, $a1 # Subtract b from a (b < a)  
b gcd # and repeat
```

.L1:

```
subu $a1, $a1, $a0 # Subtract a from b (a < b)  
b gcd # and repeat
```

.L2:

```
move $v0, $a0 # return a  
j $ra # Return to caller
```

Comments

Euclid's Algorithm in MIPS Assembly



gcd:

beq \$a0, \$a1, .L2 # if a = b, go to exit

sgt \$v0, \$a1, \$a0 # Is b > a?

bne \$v0, \$zero, .L1 # Yes, goto .L1

→ subu \$a0, \$a0, \$a1 # Subtract b from a (b < a)

b gcd # and repeat

$${}^b \cancel{\$21} = \cancel{\$21} - {}^a \cancel{\$20}$$

.L1:

subu \$a1, \$a1, \$a0 # Subtract a from b (a < b)

b gcd # and repeat

.L2:

move \$v0, \$a0 # return a

j \$ra # Return to caller

Arithmetic Instructions

Euclid's Algorithm in MIPS Assembly

gcd:

Branch if equal
beq \$a0, \$a1, .L2 # if a = b, go to exit

sgt \$v0, \$a1, \$a0 # Is b > a?

bne \$v0, \$zero, .L1 # Yes, goto .L1

Uncond branch
subu \$a0, \$a0, \$a1 # Subtract b from a (b < a)

b gcd # and repeat

.L1:

subu \$a1, \$a1, \$a0 # Subtract a from b (a < b)

b gcd # and repeat

.L2:

move \$v0, \$a0 # return a

j \$ra # Return to caller

Control-transfer instructions

General-Purpose Registers

Name	Number	Usage	Preserved?
\$zero	0	Constant zero = 0	
→ \$at	1	Reserved (assembler) <i>Don't touch</i>	
\$v0-\$v1	2-3	Function result	
↖ \$a0-\$a3	4-7	Function arguments	
\$t0-\$t7	8-15	Temporaries	
\$s0-\$s7	16-23	Saved ←	yes
\$t8-\$t9	24-25	Temporaries X	
\$k0-\$k1	26-27	Reserved (OS) <i>Don't touch</i>	
\$gp	28	Global pointer	yes
\$sp	29	Stack pointer	yes
\$fp	30	Frame pointer	yes
→ \$ra	31	Return address	yes

Each 32 bits wide

Only 0 truly behaves differently; usage is convention

Types of Instructions



Computational

Arithmetic and logical operations



Load and Store

Writing and reading data to/from memory



Jump and branch

Control transfer, often conditional



Miscellaneous

Everything else

Computational Instructions

Arithmetic

add	Add	$\$1, \$2, \$3$
addu	Add unsigned	
sub	Subtract	
subu	Subtract unsigned	?
slt	Set on less than	$a < b$
sltu	Set on less than unsigned	
and	AND	
or	OR	
xor	Exclusive OR	
nor	NOR	

Arithmetic (immediate)

addi	Add immediate	$\$1, \$2, 42$
addiu	Add immediate unsigned	
slti	Set on l. t. immediate	
sltiu	Set on less than unsigned	
andi	AND immediate	
ori	OR immediate	
xori	Exclusive OR immediate	
lui	Load upper immediate	

Shift Instructions

sll	Shift left <u>logical</u>	←
srl	Shift right logical	→
sra	Shift right <u>arithmetic</u>	sign bit
sllv	Shift left logical variable	
srlv	Shift right logical variable	
srav	Shift right arith. variable	

Multiply/Divide

mult	Multiply	}
multu	Multiply unsigned	
div	Divide	} ←
divu	Divide unsigned	
mfhi	Move from HI	
mthi	Move to HI	
mflo	Move from LO	}
mtlo	Move to LO	

$$\underbrace{\$1}_{32} \times \underbrace{\$2}_{32} = \underbrace{\text{HI LO}}_{64}$$

Computational Instructions

Arithmetic, logical, and other computations. Example:

`add $t0, $t1, $t3`
\$t0 = \$t1 + \$t3

"Add the contents of registers \$t1 and \$t3; store the result in \$t0"

Register form:

add \$t0, \$t0, \$t1

operation R_D , R_S , R_T
source "target"
source

"Perform *operation* on the contents of registers R_S and R_T ; store the result in R_D "

Passes control to the next instruction in memory after running.

Arithmetic Instruction Example

a	b	c	f	g	h	i	j
\$s0	\$s1	\$s2	\$s3	\$s4	\$s5	\$s6	\$s7

a = b - c;
f = (g + h) - (i + j);

temp → **subu** \$s0, \$s1, \$s2
addu \$t0, \$s4, \$s5 *g+h*
addu \$t1, \$s6, \$s7 *i+j*
subu \$s3, \$t0, \$t1 *f*

“Signed” addition/subtraction (**add/sub**) throw an exception on a two’s-complement overflow; “Unsigned” variants (**addu/subu**) do not. Resulting bit patterns identical.

Use these

Bitwise Logical Operator Example

11111111 | 00000000

FF00
32-bit constant

```
li $t0, 0xFF00FF00 # "Load immediate"  
li $t1, 0xF0F0F0F0 # "Load immediate"
```

nor \$t2, \$t0, \$t1 # Puts 0x000F000F in \$t2

FF00
000F000F

```
li $v0, 1 # print_int  
move $a0, $t2 # print contents of $t2  
syscall
```

= print_int

Print F000F

Immediate Computational Instructions

Example:

$$\$t0 = \$t1 + 42$$

addiu \$t0, \$t1, 42

“Add the contents of register \$t1 and 42; store the result in register \$t0”

In general,

operation R_D, R_S, I 16-bit constant

“Perform *operation* on the contents of register R_S and the signed 16-bit immediate I ; store the result in R_D ”

Thus, I can range from -32768 to 32767 . ←

32-Bit Constants and lui

It is easy to load a register with a constant from -32768 to 32767, e.g.,

```
ori $t0, $0, 42
```

Larger numbers use “load upper immediate,” which fills a register with a 16-bit immediate value followed by 16 zeros; an OR handy fills in the rest. E.g., Load \$t0 with 0xCODEFACE:

```
lui $t0, 0xCODE
```

```
ori $t0, $t0, 0xFACE
```

\$t0 = CODE0000

\$t0 = CODEFACE

The assembler automatically expands the **li** pseudo-instruction into such an instruction sequence

```
li $t1, 0xCAFE0BOE →
```

```
lui $t1, 0xCAFE
```

```
ori $t1, $t1, 0x0BOE
```

ori \$t1, \$t1, const

Multiplication and Division

Multiplication gives 64-bit result in two 32-bit registers: HI and LO. Division: LO has quotient; HI has remainder.

```
int multdiv(  
    int a,          // $a0  
    int b,          // $a1  
    unsigned c,    // $a2  
    unsigned d) // $a3  
{  
    a = a * b + c;  
    c = c * d + a;  
  
    a = a / c;  
    b = b % a;  
    c = c / d;  
    d = d % c;  
  
    return a + b + c + d;  
}
```

Handwritten annotations and assembly code:

32 HI | 32 LO

2x b

+ c

1st

2x (b+c)

2nd


```
multdiv:  
    mult $a0,$a1      # a * b (HI,LO) = $20x40  
    mflo $t0          # a * b  
    addu $a0,$t0,$a2 # a = a*b + c  
    mult $a2,$a3      # c * d  
    mflo $t1          # c * d  
    addu $a2,$t1,$a0 # c = c*d + a  
    divu $a0,$a2      # a / c  
    mflo $a0          # a = a/c  
    div $0,$a1,$a0    # b % a  
    mfhi $a1          # b = b%a  
    divu $a2,$a3      # c / d  
    mflo $a2          # c = c/d  
    addu $t2,$a0,$a1 # a + b  
    addu $t2,$t2,$a2 # (a+b) + c  
    divu $a3,$a2      # d % c  
    mfhi $a3          # d = d%c  
    addu $v0,$t2,$a3 # ((a+b)+c) + d  
j $ra
```

Shift Left

Shifting left amounts to multiplying by a power of two. Zeros are added to the least significant bits. The constant form explicitly specifies the number of bits to shift:

```
sll $a0, $a0, 1
```

1 bits to shift by



The variable form takes the number of bits to shift from a register (mod 32):

```
sllv $a1, $a0, $t0
```


lower 5 bits
of bits

Shift Right Logical

unsigned binary

The logical form of right shift adds 0's to the MSB.


```
sr1 $a0, $a0, 1
```



Shift Right Arithmetic

The “arithmetic” form of right shift sign-extends the word by copying the MSB.

```
sra $a0, $a0, 2
```



Set on Less Than

signed

$\$t1 < \$t2$?

= true

slt \$t0, \$t1, \$t2

= false

Set \$t0 to 1 if the contents of \$t1 < \$t2; 0 otherwise. \$t1 and \$t2 are treated as 32-bit signed two's complement numbers.

```
int compare(int a,          // $a0
            int b,          // $a1
            unsigned c,    // $a2
            unsigned d)    // $a3
{
    int r = 0;             // $v0
    if (a < b) r += 42;
    if (c < d) r += 99;
    return r;
}

compare:
    move $v0, $zero
    slt  $t0, $a0, $a1
    beq  $t0, $zero, .L1
    addi $v0, $v0, 42
.L1:
    sltu $t0, $a2, $a3
    beq  $t0, $zero, .L2
    addi $v0, $v0, 99
.L2:
    j    $ra
```

unsigned

Load and Store Instructions

Load/Store Instructions

	lb	Load byte	
Load	lbu	Load byte unsigned	Byte
	lh	Load halfword	
	lhu	Load halfword unsigned	Halfword
	lw	Load word	= 16 bits word
	lwl	Load word left	
	lwr	Load word right	
Store	sb	Store byte	
	sh	Store halfword	
	sw	Store word	Address multiple of 4
	swl	Store word left	
	swr	Store word right	

The MIPS is a load/store architecture: data must be moved into registers for computation.

Other architectures e.g., (x86) allow arithmetic directly on data in memory.

Memory on the MIPS

Byte
Address

Memory is byte-addressed.
Each byte consists of eight bits:

7	6	5	4	3	2	1	0
---	---	---	---	---	---	---	---



Bytes have non-negative integer addresses. Byte addresses on the 32-bit MIPS processor are 32 bits; 64-bit processors usually have 64-bit addresses.

0:	7	6	5	4	3	2	1	0
1:	7	6	5	4	3	2	1	0
2:	7	6	5	4	3	2	1	0



← 2nd byte


⋮

$2^{32} - 1$:	7	6	5	4	3	2	1	0
----------------	---	---	---	---	---	---	---	---

4 Gb total

Base Addressing in MIPS

There is only one way to refer to what address to load/store in MIPS: base + offset.
16-bit literal/constant
contents of reg.



Byte Load and Store

MIPS registers are 32 bits (4 bytes). Loading a byte into a register either clears the top three bytes or sign-extends them.

42: **F0**
↓
lbu \$t0, 42(\$0)

\$t0: **000000F0**

unsigned = 0 extend

42: **F0**
↓
lb \$t0, 42(\$0)

\$t0: **FFFFFFF0**

signed = sign-extended

The Endian Question

MIPS can also load and store 4-byte words and 2-byte halfwords.

The *endian* question: when you read a word, in what order do the bytes appear?

Little Endian: Intel, DEC, et al.


Big Endian: Motorola, IBM, Sun, et al.

MIPS can do either


SPIM adopts its host's convention

little-endian
(Intel)

Big Endian




Little Endian



Testing Endianness

```
.data                                # Directive: "this is data"
myword:                               # Define a word of data (=0)
    .word 0
.text                                  # Directive: "this is program"
main:
    la $t1, myword                    # pseudoinstruction: load address
    li $t0, 0x11
    sb $t0, 0($t1)                    # Store 0x11 at byte 0
    li $t0, 0x22
    sb $t0, 1($t1)                    # Store 0x22 at byte 1
    li $t0, 0x33
    sb $t0, 2($t1)                    # Store 0x33 at byte 2
    li $t0, 0x44
    sb $t0, 3($t1)                    # Store 0x44 at byte 3
    lw $t2, 0($t1)                    # 0x11223344 or 0x44332211?
    j $ra
```



Alignment

Word and half-word loads and stores must be *aligned*: words must start at a multiple of 4 bytes; halfwords on a multiple of 2.

Byte load/store has no such constraint.

must be multiple of 4

`lw $t0, 4($0) # OK ✓`
`lw $t0, 5($0) # BAD: $5 \bmod 4 = 1$ ✗`
`lw $t0, 8($0) # OK ✓`
`lw $t0, 12($0) # OK ✓`


`lh $t0, 2($0) # OK ✓`
`lh $t0, 3($0) # BAD: $3 \bmod 2 = 1$ ✗`
`lh $t0, 4($0) # OK ✓`

must be even


Jump and Branch Instructions

Jump and Branch Instructions

j	Jump
jal	Jump and link
jr	Jump to register
jalr	Jump and link register
beq	Branch on equal
bne	Branch on not equal
blez	Branch on less than or equal to zero
bgtz	Branch on greater than zero
bltz	Branch on less than zero
bgez	Branch on greater than or equal to zero
bltzal	Branch on less than zero and link
bgezal	Branch on greter than or equal to zero and link



Jumps



MIPS instruction = 32 bits

addresses multiples of 4

The simplest form,

①
② **j** mylabel
...

lower 2 = 00
26 bits

mylabel:

← instruction label

③ # ...

sends control to the instruction at *mylabel*. Instruction holds a 26-bit constant multiplied by four; top four bits come from current PC. Uncommon.

Jump to register sends control to a 32-bit absolute address in a register:

jr \$t3

← 32 bit address ←

Instructions must be four-byte aligned;
the contents of the register must be a multiple of 4.


Jump and Link

Jump and link stores a return address in `$ra` for implementing subroutines:

①
② `jal mysub` — *new PC* — `$ra`
⑤ # ← Control resumes here after the `jr`
⑥ # ...

`mysub:`
③ # ...
④ `jr $ra` # Jump back to caller

`jalr` is similar; target address supplied in a register.



Messes with \$t0

Branches

Used for conditionals or loops. E.g., “send control to *myloop* if the contents of \$t0 is not equal to the contents of \$t1.”

myloop:

...

3
 $\$t0 \neq \$t1$

bne \$t0, \$t1, myloop

...

16-bit offset

-82768 instructions

32767 instr.

beq is similar “branch if equal”

A “jump” supplies an absolute address; a “branch” supplies an offset to the program counter.

On the MIPS, a 16-bit signed offset is multiplied by four and added to the address of the next instruction.

Branches

Another family of branches tests a single register:

```
bgez $t0, myelse # Branch if $t0 positive  
# ...
```

myelse:

```
# ...
```

Others in this family:

blez Branch on less than or equal to zero

bgtz Branch on greater than zero

bltz Branch on less than zero

✗ **bltzal** Branch on less than zero and link

bgez Branch on greater than or equal to zero

✗ **bgezal** Branch on greater than or equal to zero and link

“and link” variants also (always) put the address of the next instruction into \$ra, just like **jal**.

Other Instructions

syscall causes a system call exception, which the OS catches, interprets, and usually returns from.

SPIM provides simple services: printing and reading integers, strings, and floating-point numbers, sbrk() (memory request), and exit().

```
# prints "the answer = 5"
```

```
.data
```

```
str:
```

```
.asciiz "the answer = "0
```

```
.text
```

```
li $v0, 4 # system call code for print_str
```

```
la $a0, str # address of string to print
```

```
syscall # print the string
```

```
li $v0, 1 # system call code for print_int
```

```
li $a0, 5 # integer to print
```

```
syscall # print it
```

This is data

the answer = 5

This is program

Other Instructions

Exception Instructions

tge tlt ...	Conditional traps	} send control to the OS
break	Breakpoint trap, for debugging	
eret	Return from exception	

Multiprocessor Instructions

ll sc Atomic	Load linked/store conditional for atomic operations
sync	Read/Write fence: wait for all memory loads/stores

Coprocessor 0 Instructions (System Mgmt)

lwr lwl ...	Cache control
tlbr tblwi ...	TLB control (virtual memory)
...	Many others (data movement, branches)

Floating-point Coprocessor Instructions

add.d sub.d ...	Arithmetic and other functions	✓
lwc1 swc1 ...	Load/store to (32) floating-point registers	
bct1t ...	Conditional branches	

Instruction Encoding

32-bit

add \$d, \$s, \$t


Register-type: add, sub, xor, ...



one of 32

Immediate-type: addi, subi, beq, ...

addi \$t, \$s, 42




addi or bne

Immediate value

16-bits

Jump-type: j, jal ...




j or jal

Register-type Encoding Example





`add $t0, $s1, $s2` *3 reg.*

add encoding from the MIPS instruction set reference:



Since \$t0 is register **8**, \$s1 is **17**, and \$s2 is **18**,



Register-type Shift Instructions




`sra $t0, $s1, 5`

sra encoding from the MIPS instruction set reference:



Since \$t0 is register 8 and \$s1 is 17,



Immediate-type Encoding Example

op:6	rs:5	rt:5	imm:16
------	------	------	--------

`addiu $t0, $s1, -42`

`addiu` encoding from the MIPS instruction set reference:


ADDIU 001001	rs	rt	immediate
-----------------	----	----	-----------

Since `$t0` is register 8 and `$s1` is 17,

<u>001001</u>	¹⁷ 10001	⁸ 01000	1111 1111 1101 0110
---------------	---------------------	--------------------	---------------------

16-bit

Jump-Type Encoding Example




jal 0x5014


multiple of 4

... 000 0101 0000 0001 0100

jal encoding from the MIPS instruction set reference:



Instruction index is a word address



16 bits word-aligned

00

Assembler Pseudoinstructions

Branch always	b <i>label</i>	→ beq \$0, \$0, <i>label</i>
Branch if equal zero	beqz <i>s</i> , <i>label</i>	→ beq <i>s</i> , \$0, <i>label</i>
Branch greater or equal	bge <i>s</i> , <i>t</i> , <i>label</i>	→ slt \$1, <i>s</i> , <i>t</i> beq \$1, \$0, <i>label</i>
Branch greater or equal unsigned	bgeu <i>s</i> , <i>t</i> , <i>label</i>	→ sltu \$1, <i>s</i> , <i>t</i> beq \$1, \$0, <i>label</i>
Branch greater than	bgt <i>s</i> , <i>t</i> , <i>label</i>	→ slt \$1, <i>t</i> , <i>s</i> bne \$1, \$0, <i>label</i>
Branch greater than unsigned	bgtu <i>s</i> , <i>t</i> , <i>label</i>	→ sltu \$1, <i>t</i> , <i>s</i> bne \$1, \$0, <i>label</i>
Branch less than	blt <i>s</i> , <i>t</i> , <i>label</i>	→ slt \$1, <i>s</i> , <i>t</i> bne \$1, \$0, <i>label</i>
Branch less than unsigned	bltu <i>s</i> , <i>t</i> , <i>label</i>	→ sltu \$1, <i>s</i> , <i>t</i> bne \$1, \$0, <i>label</i>

Assembler Pseudoinstructions

Load immediate $0 \leq j \leq 65535$	<code>li d, j</code>	<code>→ ori d, \$0, j</code>	<i>binary, not two's complement</i>
Load immediate $-32768 \leq j < 0$	<code>li d, j</code>	<code>→ addiu d, \$0, j</code>	
Load immediate	<code>li d, j</code>	<code>→ liu d, hi16(j) ori d, d, lo16(j)</code>	
Move	<code>move d, s</code>	<code>→ or d, s, \$0</code>	
Multiply	<i>if the result will fit in 32 bits</i> <code>mul d, s, t</code>	<code>→ mult s, t mflo d</code>	<i>s x t → (H, L) LO → d</i>
Negate unsigned	<code>negu d, s</code>	<code>→ subu d, \$0, s</code>	<i>0 - s</i>
Set if equal	<code>seq d, s, t</code>	<code>→ xor d, s, t sltiu d, d, 1</code>	
Set if greater or equal	<code>sge d, s, t</code>	<code>→ <u>slt d, s, t</u> xori d, d, 1</code>	<i>complement bit 0</i>
Set if greater or equal unsigned	<code>sgeu d, s, t</code>	<code>→ sltu d, s, t xori d, d, 1</code>	
Set if greater than	<code>sgt d, s, t</code>	<code>→ slt d, t, s</code>	

Expressions

Initial expression:

$$x + y + z * (w + 3)$$

Reordered to minimize intermediate results; fully parenthesized to make order of operation clear.

$$(((w + 3) * z) + y) + x$$

Assumption

addiu $\text{\$t0}$, $\text{\$a0}$, 3

mul $\text{\$t0}$, $\text{\$t0}$, $\text{\$a3}$

addu $\text{\$t0}$, $\text{\$t0}$, $\text{\$a2}$

addu $\text{\$t0}$, $\text{\$t0}$, $\text{\$a1}$

w: $\text{\$a0}$

x: $\text{\$a1}$

y: $\text{\$a2}$

z: $\text{\$a3}$

$w + 3$

$(w + 3) * z$

$+ y$

$+ x \dots$

result

Consider an alternative:

$$(x + y) + ((w + 3) * z)$$

$a + b$
A

addu $\text{\$t0}$, $\text{\$a1}$, $\text{\$a2}$

addiu $\text{\$t1}$, $\text{\$a0}$, 3

mul $\text{\$t1}$, $\text{\$t1}$, $\text{\$a3}$

addu $\text{\$t0}$, $\text{\$t0}$, $\text{\$t1}$

$x + y$

Need a second temporary

$w + 3$

$(w + 3) * z$

$(x + y) + (w + 3) * z$

Conditionals

①
if ((x + y) < 3)

x = x + 5;

else

y = y + 4;

addu \$t0, \$a0, \$a1 # x + y

slti \$t0, \$t0, 3 # (x+y) < 3

beq \$t0, \$0, ELSE

addiu \$a0, \$a0, 5 # x += 5

b DONE

ELSE:

addiu \$a1, \$a1, 4 # y += 4

DONE:

test
not
beq to 0
THEN
b done
else
else
done

zr vf \$t0 < 3
is else x+y

then branch

Do-While Loops

add
addu → can throw an exception

Post-test loop: body always executes once

```
a = 0;  
b = 0;  
do {  
  a = a + b;  
  b = b + 1;  
} while (b != 10);
```

outside

→ move \$a0, \$0 # a = 0

move \$a1, \$0 # b = 0

→ li \$t0, 10 # load constant

TOP:

addu \$a0, \$a0, \$a1 # a = a + b

addiu \$a1, \$a1, 1 # b = b + 1

bne \$a1, \$t0, TOP # b != 10?

"b"

16 bit field

While Loops

Pre-test loop: body may never execute

Tested first

```
a = 0; | move $a0, $0 # a = 0
b = 0; | move $a1, $0 # b = 0
while (b != 10) { | li $t0, 10
    a = a + b; | b TEST # test first
    b = b + 1; | BODY:
} | || addu $a0, $a0, $a1 # a = a + b
| || addiu $a1, $a1, 1 # b = b + 1
| TEST:
| bne $a1, $t0, BODY # b != 10?
```

For Loops

"Syntactic sugar" for a while loop

```
for (a = b = 0 ; b != 10 ; b++)
```

```
  a += b;
```

is equivalent to

```
a = b = 0;
```

```
while (b != 10) {
```

```
  a = a + b;
```

```
  b = b + 1;
```

```
}
```

```
move $a1, $0 # b = 0
```

```
move $a0, $a1 # a = b
```

```
li $t0, 10
```

```
b TEST # test first
```

```
BODY:
```

```
addu $a0, $a0, $a1 # a = a + b
```

```
addiu $a1, $a1, 1 # b = b + 1
```

```
TEST:
```

```
bne $a1, $t0, BODY # b != 10?
```

Arrays

```
int a[5];
```

"global variable"

5 * 4

```
void main() {  
    a[4] = a[3] = a[2] =  
        a[1] = a[0] = 3;  
    a[1] = a[2] * 4;  
    a[3] = a[4] * 2;  
}
```

```
.comm a, 20 # Allocate 20  
.text # Program next
```

```
main:
```

```
la $t0, a # Address of a
```

```
li $t1, 3
```

```
sw $t1, 0($t0) # a[0]
```

```
sw $t1, 4($t0) # a[1]
```

```
sw $t1, 8($t0) # a[2]
```

```
sw $t1, 12($t0) # a[3]
```

```
sw $t1, 16($t0) # a[4]
```

init
array

```
lw $t1, 8($t0) # a[2]
```

```
sll $t1, $t1, 2 # * 4
```

```
sw $t1, 4($t0) # a[1]
```

```
lw $t1, 16($t0) # a[4]
```

```
sll $t1, $t1, 1 # * 2
```

```
sw $t1, 12($t0) # a[3]
```

```
jr $ra
```

	⋮
0x10010010:	a[4]
0x1001000C:	a[3]
0x10010008:	a[2]
0x10010004:	a[1]
0x10010000:	a[0]
	⋮

Summing the contents of an array

```
int i, s, a[10];  
for (s = i = 0 ; i < 10 ; i++)  
    s = s + a[i];
```

```
move $a1, $0 # i = 0    $a1  
move $a0, $a1 # s = 0   $a0  
li $t0, 10 - constant  
la $t1, a # base address of array  
b TEST
```

BODY:

```
sll $t3, $a1, 2 # i * 4  
addu $t3, $t1, $t3 # &a[i]  
lw $t3, 0($t3) # fetch a[i]  
addu $a0, $a0, $t3 # s += a[i]  
addiu $a1, $a1, 1 i = i + 1
```

TEST:

```
sltu $t2, $a1, $t0 # i < 10?  
bne $t2, $0, BODY
```

Summing the contents of an array

```
int s, *i, a[10];  
for (s=0, i = a+9 ; i >= a ; i--)  
    s += *i;
```

```
move $a0, $0          # s = 0  
la   $t0, a           # &a[0]  
addiu $t1, $t0, 36   # i = a + 9  
b    TEST
```

BODY:

```
lw   $t2, 0($t1)     # *i  
addu $a0, $a0, $t2   # s += *i  
addiu $t1, $t1, -4   # i--
```

TEST:

```
sltu $t2, $t1, $t0  # i < a  
beq  $t2, $0, BODY
```

int

update pointer

s = 0

\$t1 - pointer into array

9 x 4

pointer 2

Strings: Hello World in SPIM

```
# For SPIM: "Enable Mapped I/O" must be set
# under Simulator/Settings/MIPS
.data
hello:
    .asciiz "Hello World!\n"

.text
main:
    la    $t1, 0xffff0000 # I/O base address
    la    $t0, hello
wait:
    lw    $t2, 8($t1)     # Read Transmitter control
    andi  $t2, $t2, 0x1   # Test ready bit
    beq   $t2, $0, wait

    lbu   $t2, 0($t0)     # Read the byte
    beq   $t2, $0, done   # Check for terminating 0

    sw    $t2, 12($t1)    # Write transmit data

    addiu $t0, $t0, 1     # Advance to next character
    b     wait
done:
    jr    $ra
```

Hello World in SPIM: Memory contents

```
[00400024] 3c09ffff lui   $9, -1
[00400028] 3c081001 lui   $8, 4097 [hello]
[0040002c] 8d2a0008 lw    $10, 8($9)
[00400030] 314a0001 andi  $10, $10, 1
[00400034] 1140fffe beq   $10, $0, -8 [wait]
[00400038] 910a0000 lbu   $10, 0($8)
[0040003c] 11400004 beq   $10, $0, 16 [done]
[00400040] ad2a000c sw    $10, 12($9)
[00400044] 25080001 addiu $8, $8, 1
[00400048] 0401fff9 bgez  $0 -28 [wait]
[0040004c] 03e00008 jr    $31
```

```
[10010000] 6c6c6548 6f57206f H e l l o   W o
[10010008] 21646c72 0000000a r l d ! . . . .
```


ASCII

	0	1	2	3	4	5	6	7
0:	NUL '\0'	DLE		0	@	P	'	p
1:	SOH	DC1	!	1	A	Q	a	q
2:	STX	DC2	"	2	B	R	b	r
3:	ETX	DC3	#	3	C	S	c	s
4:	EOT	DC4	\$	4	D	T	d	t
5:	ENQ	NAK	%	5	E	U	e	u
6:	ACK	SYN	&	6	F	V	f	v
7:	BEL '\a'	ETB	'	7	G	W	g	w
8:	BS '\b'	CAN	(8	H	X	h	x
9:	HT '\t'	EM)	9	I	Y	i	y
A:	LF '\n'	SUB	*	:	J	Z	j	z
B:	VT '\v'	ESC	+	;	K	[k	{
C:	FF '\f'	FS	,	<	L	\	l	
D:	CR '\r'	GS	-	=	M]	m	}
E:	SO	RS	.	>	N	^	n	~
F:	SI	US	/	?	O	_	o	DEL

Subroutines

a.k.a. procedures, functions, methods, et al.

Code that can run then resume whatever invoked it.

Exist for three reasons:

- ▶ **Code reuse**
Recurring computations aside from loops
Function libraries
- ▶ **Isolation/Abstraction**
Think Vegas:
What happens in a function stays in the function.
- ▶ **Enabling Recursion**
Fundamental to divide-and-conquer algorithms

Calling Conventions

\$t0

\$s0

Call mysub: args in \$a0,...,\$a3

saved jal mysub ← *Link args to return here*

Control returns here

Return value in \$v0 & \$v1

~~\$s0,...,\$s7, \$gp, \$sp, \$fp, \$ra~~ unchanged

\$t0 ?

\$s0 ✓

\$a0,...,\$a3, \$t0,...,\$t9 possibly clobbered

mysub: # Entry point: \$ra holds return address ←

First four args in \$a0, \$a1, ..., \$a3

... body of the subroutine ...

\$v0, and possibly \$v1, hold the result

\$s0,...,\$s7 restored to value on entry


\$gp, \$sp, \$fp, and \$ra also restored

jr \$ra # Return to the caller ←

Return address



The Stack




Towers of Hanoi



```
void move(int src, int tmp,
          int dst, int n)
{
    if (n) {
        move(src, dst, tmp, n-1);
        printf("%d->%d\n", src, dst);
        move(tmp, src, dst, n-1);
    }
}
```

Annotations: Red arrows and boxes highlight recursive calls and return values. A red arrow labeled "Do Work" points to the printf statement.



hmove:

```
addiu $sp, $sp, -24
beq   $a3, $0, L1
sw    $ra, 0($sp)
sw    $s0, 4($sp)
sw    $s1, 8($sp)
sw    $s2, 12($sp)
sw    $s3, 16($sp)
```

multiple
of 8

24
↓

\$a0	\$a1	\$a2	\$a3
src	tmp	dst	n


Allocate 24 stack bytes:
multiple of 8 for alignment

Check whether $n == 0$

Save $\$ra, \$s0, \dots, \$s3$ on the stack

20
bytes

↳ 24



hmove:

addiu \$sp, \$sp, -24

beq \$a3, \$0, L1

sw \$ra, 0(\$sp)

sw \$s0, 4(\$sp)

sw \$s1, 8(\$sp)

sw \$s2, 12(\$sp)

sw \$s3, 16(\$sp)

move \$s0, \$a0 *src*

move \$s1, \$a1 *tmp*

move \$s2, \$a2 *dst*

addiu \$s3, \$a3, -1 *n-1*

Save src in \$s0

Save tmp in \$s1

Save dst in \$s2

Save n - 1 in \$s3

hmove:

```
addiu $sp, $sp, -24
beq   $a3, $0, L1
sw    $ra, 0($sp)
sw    $s0, 4($sp)
sw    $s1, 8($sp)
sw    $s2, 12($sp)
sw    $s3, 16($sp)
```

```
move  $s0, ←$a0
move  $s1, ←$a1
move  $s2, ←$a2
addiu $s3, ←$a3, -1
```

```
move  $a1, $s2
move  $a2, $s1
move  $a3, ←$s3
```

```
jal  hmove
```

\$20 is s10

Call

hmove(src, dst, tmp, n-1)

n-1 saved all the \$s registers

→

hmove:

```
addiu $sp, $sp, -24
beq   $a3, $0, L1
sw    $ra, 0($sp)
sw    $s0, 4($sp)
sw    $s1, 8($sp)
sw    $s2, 12($sp)
sw    $s3, 16($sp)
```

```
move  $s0, $a0
move  $s1, $a1
move  $s2, $a2
addiu $s3, $a3, -1
```

```
move  $a1, $s2
move  $a2, $s1
move  $a3, $s3
jal   hmove
```

```
li    $v0, 1 # print_int
move  $a0, $s0 "src"
syscall
```

```
li    $v0, 4 # print_str " ->"
la    $a0, arrow
syscall
```

```
li    $v0, 1 # print_int
move  $a0, $s2 "dst"
syscall
li    $v0, 4 # print_str
la    $a0, newline
syscall
```

Print src -> dst

hmove:

```
addiu $sp, $sp, -24
```

```
beq $a3, $0, L1
```

```
sw $ra, 0($sp)
```

```
sw $s0, 4($sp)
```

```
sw $s1, 8($sp)
```

```
sw $s2, 12($sp)
```

```
sw $s3, 16($sp)
```

```
move $s0, $a0
```

```
move $s1, $a1
```

```
move $s2, $a2
```

```
addiu $s3, $a3, -1
```

```
move $a1, $s2
```

```
move $a2, $s1
```

```
move $a3, $s3
```

```
jal hmove
```

```
li $v0, 1 # print_int
```

```
move $a0, $s0
```

```
syscall
```

```
li $v0, 4 # print_str
```

```
la $a0, arrow
```

```
syscall
```

```
li $v0, 1 # print_int
```

```
move $a0, $s2
```

```
syscall
```

```
li $v0, 4 # print_str
```

```
la $a0, newline
```

```
syscall
```

```
move $a0, $s1
```

```
move $a1, $s0
```

```
move $a2, $s2
```

```
move $a3, $s3
```

```
jal hmove
```

Call

hmove(tmp, src, dst, n-1)

Defensive
might not
be necessary?

②

①

n-1

n-1

Recursive call

```
hmove:
    addiu $sp, $sp, -24
    beq   $a3, $0, L1
    sw   $ra, 0($sp)
    sw   $s0, 4($sp)
    sw   $s1, 8($sp)
    sw   $s2, 12($sp)
    sw   $s3, 16($sp)

    move $s0, $a0
    move $s1, $a1
    move $s2, $a2
    addiu $s3, $a3, -1

    move $a1, $s2
    move $a2, $s1
    move $a3, $s3
    jal  hmove

    li   $v0, 1 # print_int
    move $a0, $s0
    syscall
    li   $v0, 4 # print_str
    la   $a0, arrow
    syscall
```

```
li   $v0, 1 # print_int
move $a0, $s2
syscall
li   $v0, 4 # print_str
la   $a0, newline
syscall

move $a0, $s1
move $a1, $s0
move $a2, $s2
move $a3, $s3
jal  hmove

lw   $ra, 0($sp)
lw   $s0, 4($sp)
lw   $s1, 8($sp)
lw   $s2, 12($sp)
lw   $s3, 16($sp)
```

restore

Restore variables

hmove:

addiu \$sp, \$sp, -24

beq \$a3, \$0, L1

sw \$ra, 0(\$sp)

sw \$s0, 4(\$sp)

sw \$s1, 8(\$sp)

sw \$s2, 12(\$sp)

sw \$s3, 16(\$sp)

move \$s0, \$a0

move \$s1, \$a1

move \$s2, \$a2

addiu \$s3, \$a3, -1

move \$a1, \$s2

move \$a2, \$s1

move \$a3, \$s3

jal hmove

li \$v0, 1 # print_int

move \$a0, \$s0

syscall

li \$v0, 4 # print_str

la \$a0, arrow

syscall

li \$v0, 1 # print_int

move \$a0, \$s2

syscall

li \$v0, 4 # print_str

la \$a0, newline

syscall

move \$a0, \$s1

move \$a1, \$s0

move \$a2, \$s2

move \$a3, \$s3

jal hmove

lw \$ra, 0(\$sp)

lw \$s0, 4(\$sp)

lw \$s1, 8(\$sp)

lw \$s2, 12(\$sp)

lw \$s3, 16(\$sp)

L1:

addiu \$sp, \$sp, 24 # free

jr \$ra # return

.data

arrow: **.asciiz** "->"

newline: **.asciiz** "\n"

Pop 24
bytes

Factorial Example

Should be iterative

(not use the stack)

```
int fact(int n) {  
    if (n < 1) return 1;  
    else return (n * fact(n - 1));  
}
```

fact:

```
    addiu $sp, $sp, -8 # allocate 2 words on stack  
    sw    $ra, 4($sp) # save return address  
    sw    $a0, 0($sp) # and n  
    slti  $t0, $a0, 1 # n < 1?  
    beq   $t0, $0, ELSE  
    li    $v0, 1 # Yes, return 1  
    addiu $sp, $sp, 8 # Pop 2 words from stack  
    jr    $ra # return
```

ELSE:


```
    addiu $a0, $a0, -1 # No: compute n-1  
    jal   fact # recurse (result in $v0)  
    lw    $a0, 0($sp) # Restore n and  
    lw    $ra, 4($sp) # return address  
    mul   $v0, $a0, $v0 # Compute n * fact(n-1)  
    addiu $sp, $sp, 8 # Pop 2 words from stack  
    jr    $ra # return
```

n



n fact(n-1)

Memory Layout



Differences in Other ISAs

More or fewer general-purpose registers (Itanium: 128; 6502: 3)

Arithmetic instructions affect condition codes (e.g., zero, carry);
conditional branches test these flags

Registers that are more specialized (x86)

More addressing modes (x86: 6; VAX: 20)

Arithmetic instructions that also access memory (x86; VAX)

Arithmetic instructions on other data types (bytes and halfwords)

Variable-length instructions (x86; ARM)

Predicated instructions (ARM, VLIW)

RISC

Single instructions that do much more (x86 string move,
procedure entry/exit)